


# Лекция 0xС

26 марта



# Стандартная библиотека языка Си


- 24 заголовочных файла
- `stdlib.h`
  - Преобразование типов: `atoi`, `strtod`, ...
  - Генерация псевдослучайных последовательностей
  - Выделение и освобождение памяти
  - Сортировка и поиск
  - Математика
- `stdio.h`
  - Функции для файловых операций
  - Функции для операций ввода-вывода
- `string.h`
- ...

# Выравнивание стека

- На стек помещаем только машинные слова (4 байта)
- Выравнивание стека по границе в 16 байт  
IA-32/Linux/gcc
  - выполняется в функции `main`
  - остальные функции поддерживают выравнивание, формируя фрейм определенного размера
  - для листовых функций необязательно
- Не только производительность: команда `MOVDQA`
  - SSE2, Pentium4, 2001 г.
  - Быстрое копирование блока данных размером 16 байт
  - Аварийное завершение работы (`#GP`), если начальный адрес блока не выровнен по границе в 16 байт

# Отказ от указателя фрейма

- Пролог и эпилог функций вносят накладные расходы
  - Особенно заметно на небольших функциях
- Содержимое фрейма распределяется во время компиляции
  - как правило, все размещаемые внутри фрейма данные могут быть адресованы через константные смещения
- Используем для адресации только ESP
  - Порядок использования EBP – часть ABI (Application Binary Interface)
  - До 2011 года gcc не включал опцию `-fomit-frame-pointer` в общие списки оптимизаций



# Отказ от указателя фрейма

```
gcc -S -Os -fomit-frame-pointer -fno-optimize-sibling-calls -masm=intel length.c
```

```
typedef struct link link;
```

```
struct link {
    int payload;
    link* next;
};
```

```
int length(link *p) {
    return p? length(p->next) + 1 : 0;
}
```

```
length:
    sub     esp, 12
    xor     eax, eax
    mov     edx, dword [esp+16]
    test    edx, edx
    je     .L2
    mov     ecx, dword [edx+4]
    mov     [esp], ecx
    call   length
    inc     eax
.L2:
    add     esp, 12
    ret
```

# Отказ от указателя фрейма

```
gcc -S -Os -fno-omit-frame-pointer -fno
```

```
typedef struct link link;
```

```
struct link {
    int payload;
    link* next;
};
```

```
int length(link *p) {
    return p? length(p->next) + 1 : 0;
}
```

```
length:
```

```
    push    ebp
    xor     eax, eax
    mov     ebp, esp
    sub     esp, 8
    mov     edx, dword [ebp+8]
    test    edx, edx
    je     .L2
    mov     ecx, dword [edx+4]
    mov     dword [esp], ecx
    call   length
    inc     eax
.L2:
    leave
    ret
```

# Какие есть компиляторы и соглашения о вызове функций?

- Компиляторы
  - LLVM clang (open source)
    - AMD Optimizing C/C++ Compiler (AOCC)
  - GNU gcc (open source)
  - Microsoft Visual Studio vc
  - Intel icc
- еще компиляторы ...
  - PGI
  - C++Builder (Embarcadero)
  - Oracle Developer Studio  
последняя стабильная версия – 2017 год
  - Open Watcom  
последняя версия – 2010 год
- Модели памяти
  - 16, 32, 64 разряда
- Операционные системы
  - Linux (FreeBSD, ...)
  - Windows
  - Mac OS

Дополнительная техническая информация - домашняя страница Agner Fog:

<http://www.agner.org/optimize/> сборник материалов по компиляторам и оптимизации.

*Calling conventions for different C++ compilers and operating systems*



# Соглашение STDCALL

Очистку стека от аргументов вызова выполняет сама вызванная функция

```
#include <stdio.h>

__attribute__((stdcall))
int sum(int x, int y);

int main() {
    int a = 1, b = 2, c;
    c = sum(a, b);
    printf("%d\n", c);
    return 0;
}

__attribute__((stdcall))
int sum(int x, int y) {
    int t = x + y;
    return t;
}
```

```
sum:
    push    ebp
    mov     ebp, esp
    mov     eax, dword [ebp+12]
    add     eax, dword [ebp+8]
    pop     ebp
    ret     8
```

```
CMAIN:
    ; ...
    mov     dword [esp+4], 2
    mov     dword [esp], 1
    call    sum
    sub     esp, 8
    mov     dword [ebp-8], eax
    ; ...
```

При вызове `stdcall` из функции, имеющей соглашение `cdecl`, необходим код для компенсации освобождения стека.

# Соглашение FASTCALL

- Первый и второй параметры размещаются в регистрах ECX и EDX
  - Если размер параметров позволяет
- Остальные параметры – на стеке, от них стек очищает вызванная функция, как и в stdcall
- Такая форма соглашения принята в gcc и MSVC

```
typedef struct chain chain;
```

```
struct chain {
    unsigned data;
    chain *next;
};
```

```
__attribute__((fastcall)) unsigned xorEmAll(chain *p, unsigned salt) {
    if (p) {
        return p->data ^= xorEmAll(p->next, salt);
    } else {
        return salt;
    }
}
```

```
xorEmAll:
```

```
test    ecx, ecx
mov     eax, edx
je     .L6
push   ebx
mov     ebx, ecx
sub     esp, 8
mov     ecx, dword [ecx+4]
call   xorEmAll
xor     eax, dword [ebx]
mov     dword [ebx], eax
add     esp, 8
pop     ebx
.L6:
ret
```

x86-64

# Зачем переходить на 64-х разрядную архитектуру?

- Особенности полноценной 64-х разрядной процессорной архитектуры
  - АЛУ оперирует 64-х разрядными данными
  - (Большой) набор 64-х разрядных регистров общего назначения
  - 64-х разрядное (плоское) адресное пространство
- Преимущества 64-х разрядной процессорной архитектуры
  - Эффективнее (быстрее) работаем с 64-х разрядными данными
  - Реже «проливаем» содержимое регистров
  - Огромное пространство адресуемой памяти  
 $2^{64} = 16 \times 2^{30} \times 2^{30} = 16$  Эксбибайт
- Примеры полноценных 64-х разрядных архитектур
  - PowerPC, Sparc, Alpha, IA-64 (Itanium)

A red starburst graphic containing the text 'x86-64' in a white box.

# Что такое «Архитектура x86\_64»?

- В момент появления x86\_64 едва ли возможно отнести к полноценным 64-х разрядным архитектурам
  - Архитектура x86\_64 была получена очередным эволюционным расширением ISA IA-32
  - Двоичная кодировка команд IA-32 не изменилась. Работа с 64-х разрядными регистрами и данными реализована через специальные префиксы в коде операции, переопределяющие поведение процессора по умолчанию. Декодирование команд «оптимизировано» для работы с 32-х разрядными командами
  - Доступ к 64-х разрядному адресному пространству реализован через доработанный механизм сегментной памяти IA-32
- Почему AMD/Intel пошли таким путем?
  - Некоторые свойства 64-х разрядных процессоров достигаются, причем переход с 32-х разрядных процессоров получается проще
  - Сохранена работоспособность ISA IA-32, а следовательно – **всех ранее написанных для данной архитектуры программ.**


The move toward 64-bit computing for mainstream applications, will initially focus on applications that are already constrained by 32-bit memory limitations. ... Platforms based on the Intel Xeon processor with Intel EM64T are preferable for general purpose applications, such as Web and mail infrastructure, digital content creation, mechanical computer aided design, and electronic design automation; and for mixed environments in which optimized 32-bit performance remains critical.

Intel. *The 64-bit Tipping Point*. September 2004 <https://software.intel.com/sites/default/files/e1/4f/26944>

x86-64

# Архитектура x86-64

- Адресуемая память  $2^{64}$  байт (на самом деле - нет)
- Вдвое больше регистров общего назначения, размер регистров удвоился
- Векторные регистры и команды SSE
- Обращение к данным через смещение относительно счетчика команд
- Величины  $2^{64}$  кодируются только в команде MOV
  - Переслать непосредственно закодированный операнд можно только в регистр
  - В общем формате адресного кода операнда-памяти смещение осталось  $2^{32}$ . Можно задать абсолютный адрес  $2^{64}$ , но только если это пересылка данных из RAX



x86-64

# Регистры x86-64: Соглашение по использованию при вызове функций

rax	Возвращаемое значение
rbx	Сохраняется вызванной функцией
rcx	Аргумент #4
rdx	Аргумент #3
rsi	Аргумент #2
rdi	Аргумент #1
rsp	Указатель стека
rbp	Сохраняется вызванной функцией

r8	Аргумент #5
r9	Аргумент #6
r10	Сохраняется вызывающей функцией
r11	Сохраняется вызывающей функцией
r12	Сохраняется вызванной функцией
r13	Сохраняется вызванной функцией
r14	Сохраняется вызванной функцией
r15	Сохраняется вызванной функцией

A red starburst graphic with a white border containing the text 'x86-64' in black.

# Регистры x86-64

- Аргументы передаются в функцию через регистры
  - Если целочисленных параметров более 6, остальные передаются через стек
  - Регистры-аргументы могут рассматриваться как сохраненные на стороне вызывающей функции
- Все обращения к фрейму организованы через указатель стека
  - Отпадает необходимость поддерживать значения EBP/RBP
- Остальные регистры
  - 6 регистров сохраняется вызванной функцией
  - 2 регистра сохраняется вызывающей функцией
  - 1 регистр для возвращаемого значения *может рассматриваться как регистр, сохраненный на стороне вызывающей функции*
  - 1 выделенный регистр – указатель стека


x86-64

# Обмен значениями переменных long@x86-64

```
void swap_l(long *xp, long *yp) {
    long t0 = *xp;
    long t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

```
swap:
    mov    rdx,    qword [rdi]
    mov    rax,    qword [rsi]
    mov    qword [rdi],    rax
    mov    qword [rsi],    rdx
    ret
```

- Параметры передаются через регистры
  - Первый параметр (**xp**) был размещен в **rdi**, второй (**yp**) – в **rsi**
  - 64-разрядные указатели
- Никакие команды не работают со стеком (за исключением **ret**)
- Удалось полностью отказаться от использования стека
  - Все локальные данные размещены на регистрах





x86-64

# Локальные переменные в «красной зоне»

Листовая функция

```

/*
 * Обмен через локальный массив
 */
void swap_a(long *xp, long *yp) {
    volatile long loc[2];
    loc[0] = *xp;
    loc[1] = *yp;
    *xp = loc[1];
    *yp = loc[0];
}

```


swap\_a:

```

mov rax, qword [rdi]
mov qword [rsp-24], rax
mov rax, qword [rsi]
mov qword [rsp-16], rax
mov rax, qword [rsp-16]
mov qword [rdi], rax
mov rax, qword [rsp-24]
mov qword [rsi], rax
ret

```

- Обходимся без изменения указателя стека
  - Все данные размещены во «фрейме», неявно организованным под текущим указателем стека
  - Поддержка «красной зоны» явно прописана в ABI x86\_64: обработка **прерываний** не должна затрагивать содержимое «красной зоны»



x86-64

# Нелистовая функция без организации фрейма

```
/* Обмен a[i] и a[i+1] */
void swap_ele(long a[], int i) {
    swap(&a[i], &a[i+1]);
}
```

- На период работы swap уже никаких значений сохранять на регистрах не требуется
- Не требуется сохранять регистры в качестве вызванной функции
- Команда (префикс) rep используется вместо команды NOP
  - Рекомендации компании AMD...

Источники:

*Software Optimization Guide for AMD64 Processors*

<https://support.amd.com/TechDocs/25112.PDF> раздел 6.2

*Software Optimization Guide for AMD Family 17h Processors*

[https://developer.amd.com/wordpress/media/2013/12/55723\\_SOG\\_Fam\\_17h\\_Processors\\_3.00.pdf](https://developer.amd.com/wordpress/media/2013/12/55723_SOG_Fam_17h_Processors_3.00.pdf) раздел 2.8.1.3.2

```
swap_ele:
    movsx rsi, esi           ; знаковое расширение i
    lea  rax, [rdi + 8*rsi + 8] ; &a[i+1]
    lea  rdi, [rdi + 8*rsi]   ; &a[i] первый аргумент
    mov  rsi, rax            ; второй аргумент
    call swap
    rep                               ; пустая команда / НОП
    ret
```

x86-64

## Пример организации фрейма

```
long sum = 0;
/* Swap a[i] & a[i+1] */
void swap_ele_su
(long a[], int i) {
    swap(&a[i], &a[i+1]);
    sum += (a[i]*a[i+1]);
}
```

- Размещаем значения выражений `&a[i]` и `&a[i+1]` в регистрах, сохраняемых на стороне вызванной функции
- Необходимо сформировать фрейм для сохранения этих регистров

```
swap_ele_su:
    mov    [rsp-16], rbx
    mov    [rsp-8],  rbp
    sub    rsp, 16
    movsx  rax, esi
    lea   rbx, [rdi + 8*rax + 8]
    lea   rbp, [rdi + 8*rax]
    mov   rsi, rbx
    mov   rdi, rbp
    call  swap
    mov   rax, [rbx]
    imul rax, [rbp]
    add  [rel sum], rax
    mov  rbx, [rsp]
    mov  rbp, [rsp+8]
    add  rsp, 16
    ret
```

Для x86-64 может использоваться одна из четырех моделей построения кода  
`-mmodel=[small | medium | large | kernel]`

x86-64

# Как происходит работа с фреймом

```
swap_ele_su:
```

```
mov    [rsp-16], rbx    ; сохраняем rbx
mov    [rsp-8],  rbp    ; сохраняем rbp
sub    rsp, 16         ; выделяем на стеке место для фрейма
movsx  rax, esi        ; знаковое расширение i
lea    rbx, [rdi + 8*rax + 8] ; &a[i+1]
lea    rbp, [rdi + 8*rax]  ; &a[i]
mov    rsi, rbx        ; второй аргумент вызова
mov    rdi, rbp        ; первый аргумент вызова
call   swap
mov    rax, [rbx]      ; помещаем в rax a[i+1]
imul  rax, [rbp]       ; умножаем на a[i]
add   [rel sum], rax   ; прибавляем к переменной sum
                           ; адрес вычисляется как RIP + sum
mov    rbx, [rsp]      ; восстанавливаем значение rbx
mov    rbp, [rsp+8]    ; восстанавливаем значение rbp
add    rsp, 16         ; освобождаем место занятое фреймом
ret
```

x86-64

## Особенности работы с фреймом

- Выделение всего фрейма одной командой
  - Обращения к содержимому фрейма используют адресацию относительно `rsp`
  - Уменьшаем значение в указателе стека
  - Выделение памяти может выполняться не сразу, поскольку в определенных временных пределах хранить данные в «красной зоне» безопасно
- Простое освобождение фрейма
  - Увеличиваем значение в указателе стека
  - Указатель фрейма не требуется

A red starburst graphic with a white border containing the text "x86-64".

x86-64

# Промежуточные итоги

## x86-64 : организация вызова функций

- Активное использование регистров
  - Передача параметров
  - Больше регистров – больше возможностей вычислять временные значения и их повторно использовать
- Минимальное использование стека
  - Иногда удастся вообще его не использовать
  - Создание/освобождение всего фрейма
- Доступные оптимизации
  - В каком виде будет создан фрейм?
  - Как именно будет выполняться создание?

## Далее...

- **Функции**
  - Рекурсия
  - Выравнивание стека
  - *Различные соглашения о вызове функций*
    - *cdecl/stdcall/fastcall, отказ от указателя фрейма*
    - *Соглашение вызова для x86-64*
  - **Переполнение буфера, эксплуатация ошибок, механизмы защиты**
- Динамическая память
- Числа с плавающей точкой